Example nº 1
Forces and Moments on a beam

CivilFEM Manual of Advanced Examples Example nº 1 – Table of Contents

1 EXAMPLE Nº 1: FORCES AND MOMENTS ON A BEAM

1.1 AIM

This first example, which is a very simple case, corresponds to a bisupported structure of beams, with punctual and surface loads that can act in any point or area. Its aim is to familiarize the user with the accomplishment of linear combinations using the tools provided by CivilFEM.

1.2 DESCRIPTION OF THE EXAMPLE

In the structure of the figure



the surface load q represents an action that can be extended all throughout the beam or can act on a single region of it. The load Q is a movable action that can also act in any point.

The values of the parameters are:

Parameter	Value	Unit
γ	25	kN/m^3
q	10	kN/m
Q	80	kN
L_1	10,0	m
L_2	3,0	m
h	1,0	m
b	0,3	m

The loads safety factors are:

Type of load	Favorable effect (γ_F)	Unfavorable effect (γ_D)
Self weight	1,00	1,35
Overload (surface or punctual)	0	1,50

1.3 RESULTS TO BE OBTAINED

In order to simplify the calculations, only the following actions have been considered:

Law I - Self weight $\times \gamma_D$ in all the structure

Law II - Self weight $\times \gamma_F$ in all the structure

Law III - Surface load q $\times \gamma_D$ in AB

Law IV - Surface load q $\times \gamma_D$ in BC

Law V - Load Q $\times \gamma_D$ in AB (centre of the span)

Law IV - Load Q $\times \gamma_D$ in BC (end of the projection)

The following diagrams of bending moments and shear forces have been analytically calculated.



Bending moment envelope



Shear force envelope

The requested results are:

1. Obtain these laws using ANSYS+CivilFEM calculating the envelopes of the previously defined load states I, II, III, IV, V and IV.
2. Obtain these laws using ANSYS+CivilFEM calculating the envelopes of the load states I+III+V, II+III+V, I+IV+VI and II+IV+VI
3. Verify the maximum bending moment in A and B by partially filling the spans using the combinations module of CivilFEM.

1.4 ANALYTICAL RESOLUTION

1.4.1 Self weight of all the structure



Law I:

Self weight: $p = 0.30m \cdot 1.0m \cdot 25kN/m^3 = 7.5kN/m$

Self weight with safety factor: $p^* = 1.35 \cdot 7.5kN/m = 10.13kN/m$

By equilibrium conditions the following equations are obtained:

$$R_1 + R_2 = [b \cdot c \cdot \gamma \cdot \gamma_D] \cdot (L_1 + L_2) = 131.69$$

$$10 \cdot R_2 = [b \cdot c \cdot \gamma \cdot \gamma_D] \cdot \frac{(L_1 + L_2)}{2}$$

Where:

$$R_1 = 46.09 \text{ and } R_2 = 85.6$$

In the first section $0 \le x \le 10m$ we have it law of moments is:

$$M = 46.09 \cdot x - 10.13 \cdot x \cdot \frac{x}{2} = 46.09 \cdot x - 5.07 \cdot x^2$$

Therefore:

$x = 4.55$ and $M = 104.96$

$x = 9.1$ and $M = 0$

$x = 10$ and $M = -45.63$

Law II:

Self weight: $p = 0.30m \cdot 1.0m \cdot 25kN/m^3 = 7.5kN/m$

By equilibrium conditions the following equations are obtained:

$$R_1 + R_2 = [b \cdot c \cdot \gamma] \cdot (L_1 + L_2) = 97.5$$

$$10 \cdot R_2 = [b \cdot c \cdot \gamma] \cdot \frac{(L_1 + L_2)}{2}$$

Where:

$R_1 = 63.37$ and $R_2 = 34.13$

In the first span $0 \le x \le 10m$ the bending moments law is:

$$M = 3.75 \cdot x \cdot (9.1 \cdot x)$$

Therefore:

$x = 4.55$ and $M = 77.63$

$x = 9.1$ and $M = 0$

$x = 10$ and $M = -33.75$

1.4.2 Uniform surface load



Law III:

The moment at the centre of the span, with the safety factor, is

$$M_D = \left[\frac{q \cdot L_1}{2} \cdot \frac{L_1}{2} - \frac{q \cdot L_1}{2} \cdot \frac{L_1}{4}\right] \cdot \gamma_D = \frac{q \cdot L_1}{8} \cdot \gamma_D = 125 kN \cdot m \cdot 1.5 = 187.5 kN \cdot m$$

and the law is: $M(x) = 75 \cdot x - 7.5 \cdot x^2$ in AB

Law IV:

$$M(x) = -6{,}75 \cdot x$$

1.4.3 Punctual load at the centre of the span



Law V:

The equation of the bending moments law is:

$$M(x) = \begin{cases} 60x \quad \text{from} \;\; \text{A to D} \\ 60x - 120(x-5) \quad \text{from} \;\; \text{D to B} \end{cases}$$

and the moment at the centre of the span, with the safety factor, is

$$M_D = \frac{Q}{2} \cdot \frac{L_1}{2} \cdot \gamma_D = 300 \text{kN} \cdot \text{m}$$

Law IV:

The equation of the bending moments law is $M(x) = 36x$ and the moment at the centre of the span takes the value $M_D = 180 \text{ kN.m}$

1.4.4 Composition of bending moments

The composition of positive bending moments (I+III+V) in the first span is defined by the expression:

$$M(x) = 5.07 \cdot x \cdot (9.1 - x) + 75 \cdot x - 7.5 \cdot x^2 + 60x$$

$$M(x) = 241.14 \cdot x + 24.57 \cdot x^2$$

Obtaining for x=5m, the value $M = 591.32 \text{ m.T}$

The composition of negative bending moments that provides the minimum value turns out to be I+IV+VI

$$M(x) = 5.07 \cdot x \cdot (9.1 - x) - 6.75 \cdot x - 36 \cdot x$$

$$M(x) = 3.39 \cdot x - 5.07 \cdot x^2$$

and of this law it is deduced that for x=10m it is $M = -473.1 \text{ m.T}$

1.5 CALCULATION LOG

1.5.1 Introduction

Firstly all the parameters that will be used to create the geometry of the model are defined. The model is then created, without Boolean operations and with a down to up construction, which allows a completely parametrical geometry.

The finite elements model is created starting with a group of Keypoints, located at points A, B and C and frome these, Beam54 elements are created, (this element type is used because the problem is 2D).

Next, the beam cross section is created and the boundary conditions for the model are set (supports and loads).

The model is solved using two types of analysis.

The first one consists in solving the simple cases exposed in chapter 1.3, and combine them linearly to obtain states I+III+V, II+III+V, I+IV+VI and II+IV+VI.

The second method of analysis consists in solving in first place the self weight load case. Then the surface load is distributed in each one of the elements of the beam. And to finalize, the punctual load is located in each of the key points of the model. The combinations module of CivilFEM will chooses the worst possible combination of the loads, setting as target the maximum bending moment Z.

For this, a combination is created (5, compatible), from all the generated *data sets* when solving the surface load on each one of the elements of the beam. The combination 6, that represents the punctual load, is defined as incompatible because only one punctual load can act at a time. Combination 7 is a compatible additon of the two previous combinations. And finally combination 8 is the addition of the previous combination with the self weight.

In order to finish the example it is requested to the program to draw the bending moments and the shear forces.

1.5.2 Generation of the model

```
 FINISH
 ~CFCLEAR,,1

 /TITLE, Example 1, Beam envelope

! Units
 ~UNITS,,LENG,M
 ~UNITS,,TIME,S
 ~UNITS,,FORC,KN

! Parameters
  g  = 25
  QU = 10
  Q  = 80
  L1 = 10
```

```
  L2 = 3
  B  = 0.3
  H  = 1
! Load coefficients
  CSWF = 1.35   ! Self Weight Favorable
  CSWN = 1      ! Self Weight Non-Favorable
  CSLF = 0      ! Suface Load Favorable
  CSLN = 1.5    ! Suface Load Non-Favorable
! ----------------------------------------------------

/PREP7
! Materials
  ~CFMP,1,LIB,CONCRETE,EC2,C20/25
! Elements types
  ET,1,Beam54
! Sections
  ~CSECDMS,1,REC,1,H,B
! Beam & Shell properties
  ~BMSHPRO,1,BEAM,1,1,,,54,1,0,
! Geometry & mesh
! Node definition.
  N,1,0
  N,50,L1
  FILL
  N,65,L1+L2
  FILL
! Element definition
  E,1,2
  EGEN,64,1,-1
! Boundary conditions
  D,NODE(0,0,0),UX,0
  D,NODE(0,0,0),UY,0
  D,NODE(L1,0,0),UY,0

  EPLOT
  ALLSEL
```

1.5.3 Solution

```
! Solve

! First method
! We will solve six actions
/SOLU
! Case I: Self Weight x CSWF in all the beam
/TITLE, Self Weight x CSWF
  ESLN,S,1
  SFBEAM,ALL,1,PRES,B*H*g*CSWF
  ALLSEL,ALL
  SOLVE
  SFEDELE,ALL,ALL,ALL
! Case II: Self Weight x CSWN in all the beam
/TITLE, Self Weight x CSWN
  ESLN,S,1
  SFBEAM,ALL,1,PRES,B*H*g*CSWN
  ALLSEL,ALL
  SOLVE
  SFEDELE,ALL,ALL,ALL
! Case III: Surface Load in the span x CSLN
/TITLE,Surface Load 1 span x CSLN
  NSEL,S,LOC,X,0,L1
  ESLN,S,1
  SFBEAM,ALL,1,PRES,QU*CSLN
  ALLSEL,ALL
  SOLVE
```

```
  SFEDELE,ALL,ALL,ALL
! Case IV: Surface Load in the flange x CSLN
/TITLE,Surface Load 2 span x CSLN
  NSEL,S,LOC,X,L1,L1+L2
  ESLN,S,1
  SFBEAM,ALL,1,PRES,QU*CSLN, , , , , ,
  ALLSEL,ALL
  SOLVE
  SFEDELE,ALL,ALL,ALL
! Case V: Punctual Load in the span x CSLN
/TITLE, Punctual Load 1 span x CSLN
  F,NODE(L1/2,0,0),FY,-Q*CSLN,
  SOLVE
  FDELE,ALL
! Case VI: Punctual Load in the flange x CSLN
/TITLE,Punctual Load 2 span x CSLN
  F,NODE(L1+L2,0,0),FY,-Q*CSLN,
  SOLVE
  FDELE,ALL

! Second method to solve the problem.
! In this method we will:
! - Solve the beam with the surface load in each diferent element.
! - Solve the beam with the punctual load in each node.
! - Combine the results

! Surface load solution:
! Loop on the elements
  *GET,NELEM,ELEM,,COUNT
  *DO,I,1,NELEM
    /TITLE,Surface Load, element %I%
    SFBEAM,I,1,PRES,QU*CSLN
    SOLVE
    SFEDELE,ALL,ALL,ALL
  *ENDDO
! Punctual load solution:
! Loop on the nodes
  *GET,NNODE,NODE,,COUNT
  *DO,I,1,NNODE
    /TITLE,Punctual Load, Node %I%
    F,I,FY,-Q*CSLN,
    SOLVE
    FDELE,ALL
  *ENDDO
```

1.5.4 Postprocess

```
/POST1
~CMBCLR,,,1
~TRGDEF,1,CROSS,M,Z,MAX    ! Maximun bending moment
~TRGDEF,2,CROSS,M,Z,MIN    ! Minimun bending moment
~TRGDEF,3,CROSS,F,Y,MAX    ! Maximum shear force
~TRGDEF,4,CROSS,F,Y,MIN    ! Minimum shear force

! Combination I:
/TITLE,Maximun I+II+V ~CMBDEF,1,ADD,3,0,
~STSTDEF,1,1,NSET,1,,1,1,1,1
~STSTCFT,1,1,1.000
~STSTDEF,1,2,NSET,3,,1,1,1,1
~STSTCFT,1,2,1.000
~STSTDEF,1,3,NSET,5,,1,1,1,1
~STSTCFT,1,3,1.000
! Combination II:
/TITLE,Minimun II+III+V
~CMBDEF,2,ADD,3,0,
~STSTDEF,2,1,NSET,2,,1,1,1,1
~STSTCFT,2,1,1.000
~STSTDEF,2,2,NSET,3,,1,1,1,1
~STSTCFT,2,2,1.000
~STSTDEF,2,3,NSET,5,,1,1,1,1
~STSTCFT,2,3,1.000
! Combination III:
/TITLE,I+IV+VI
~CMBDEF,3,ADD,3,0,
~STSTDEF,3,1,NSET,1,,1,1,1,1
~STSTCFT,3,1,1.000
~STSTDEF,3,2,NSET,4,,1,1,1,1
~STSTCFT,3,2,1.000
~STSTDEF,3,3,NSET,6,,1,1,1,1
~STSTCFT,3,3,1.000
! Combination IV:
/TITLE,II+IV+VI
~CMBDEF,4,ADD,3,0,
~STSTDEF,4,1,NSET,2,,1,1,1,1
~STSTCFT,4,1,1.000
~STSTDEF,4,2,NSET,4,,1,1,1,1
~STSTCFT,4,2,1.000
~STSTDEF,4,3,NSET,6,,1,1,1,1
~STSTCFT,4,3,1.000
! Surface combination:
/TITLE,Surface Combination
~CMBDEF,5,COMPATIB,67,0,
*DO,I,1,67
  ~STSTDEF,5,I,NSET,I+6,,1,1,1,1
*ENDDO
! Puntual combination:
/TITLE,Puntual Combination
```

```
~CMBDEF,6,INCOMPAT,68,0,8
*DO,I,1,68
  ~STSTDEF,6,I,NSET,I+73,1,1,1,1
*ENDDO
! Mobile combination:
/TITLE,Mobile Combination
~CMBDEF,7,COMPATIB,2,0,
~STSTDEF,7,1,CMB,5,,1,1,1,1
~STSTDEF,7,2,CMB,6,,1,1,1,1
! Mobile combination:
/TITLE,Self weight + Mobile combination
~CMBDEF,8,ADD,2,0,2
~STSTDEF,8,1,NSET,1,0,1,1,1,1
~STSTCFT,8,1,1.000
~STSTDEF,8,2,CMB,7,,1,1,1,1
~STSTCFT,8,2,1.000
~COMBINE
!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!
! Combined results
!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!
~CMBDAT,2
/PLOPTS,TITLE,OFF    ! Graphical options
! Bending moment
/ANNOT,DELE
~CFSET,,1,1,
~PLLSFOR,M,Z,-1,
/TLABEL,-0.9,-0.9,Combination I+III+V
/ANNOT,DELE
~CFSET,,3,1,
~PLLSFOR,M,Z,-1,
/TLABEL,-0.9,-0.9,Combination I+IV+VI
/ANNOT,DELE
~CFSET,,8,3,
~PLLSFOR,F,Y,1,
/TLABEL,-0.9,-0.9,Shear solution Maximun
/ANNOT,DELE
~CFSET,,8,4,
~PLLSFOR,F,Y,1,
/TLABEL,-0.9,-0.9,Shear solution Minimun
! Bending moment, with all the surface and puntual
load.
/ANNOT,DELE
~CFSET,,8,1,
~PLLSFOR,M,Z,-1,
/TLABEL,-0.9,-0.9,Complete solution maximun
/ANNOT,DELE
~CFSET,,8,2,
~PLLSFOR,M,Z,-1,
/TLABEL,-0.9,-0.9,Complete solution minimun
```



1.6 RESULTS

In order to calculate the bending moments law the load cases described in chapter 1.3 have been calculated. Adding cases I+III+V and II+III+V and combining both results, the analytically calculated maximum bending moments envelope is obtained (in figure 1 the laws of both cases are represented). In the same way, adding cases I+IV+VI and II+IV+VI and combining both results, the minimum bending moments envelope is obtained (in figure 2 the laws of both cases are represented).

In order to calculate the shear forces envelope, all the possible locations of the surface and punctual loads are solved. Next they are combined searching for the maximum and minimum shear force. This is represented in figure 3 and figure 4. As it is shown there is a small difference with the analytical solution, due to the discretization of the model.

Flectores moments:

Cases I+III+V and II+III+V:



Figure 1

Cases I+IV+VI and II+IV+VI:



Figure 2

Sharp efforts:

Maximum:



Figure 3

Minimum:



Figure 4

If the combinations module of CivilFEM is used to choose the more unfavourable load for the bending moment, it will combine the obtained results to move the surface and punctual loads all throughout the structure. The obtained result is represented in figures 5 and 6.



Figure 5



Figure 6

1.7 SUMMARY

With this example the user can analyse the structure as the same time as he studies the possibilities of a *LOG* file in *APDL* language. It is also intended to become familiar with the combinations module, that is one of the more powerful tools of the program, and has been specially thought and designed for the necessities of Civil Engineering.